

Chapter 36

MRCHART Statement

Chapter Table of Contents

OVERVIEW	1219
GETTING STARTED	1220
Creating Charts for Medians and Ranges from Raw Data	1220
Creating Charts for Medians and Ranges from Summary Data	1222
Saving Summary Statistics	1225
Saving Control Limits	1226
Reading Preestablished Control Limits	1229
SYNTAX	1231
Summary of Options	1232
DETAILS	1242
Constructing Charts for Medians and Ranges	1242
Output Data Sets	1244
ODS Tables	1247
Input Data Sets	1248
Methods for Estimating the Standard Deviation	1251
Axis Labels	1252
Missing Values	1252
EXAMPLES	1253
Example 36.1 Working with Unequal Subgroup Sample Sizes	1253
Example 36.2 Specifying Axis Labels	1257

Chapter 36

MRCHART Statement

Overview

The MRCHART statement creates charts for subgroup medians and ranges, which are used to analyze the central tendency and variability of a process.

You can use options in the MRCHART statement to

- compute control limits from the data based on a multiple of the standard error of the plotted medians and ranges or as probability limits
- tabulate subgroup sample sizes, subgroup medians, subgroup ranges, control limits, and other information
- save control limits in an output data set
- save subgroup sample sizes, subgroup medians, and subgroup ranges in an output data set
- read preestablished control limits from a data set
- apply tests for special causes (also known as runs tests and Western Electric rules)
- specify the method for estimating the process standard deviation
- specify a known (standard) process mean and standard deviation for computing control limits
- display distinct sets of control limits for data from successive time phases
- add block legends and symbol markers to reveal stratification in process data
- superimpose stars at points to represent related multivariate factors
- clip extreme points to make the charts more readable
- display vertical and horizontal reference lines
- control axis values and labels
- control layout and appearance of the chart

Getting Started

This section introduces the MRCHART statement with simple examples that illustrate commonly used options. Complete syntax for the MRCHART statement is presented in the “Syntax” section on page 1231, and advanced examples are given in the “Examples” section on page 1253.

Creating Charts for Medians and Ranges from Raw Data

See SHWMR1
 in the SAS/QC
 Sample Library

A consumer products company weighs detergent boxes (in pounds) to determine whether the fill process is in control. The following statements create a SAS data set named DETERGNT, which contains the weights for five boxes in each of 28 lots. A lot is considered a rational subgroup.

```

data detergent;
  input lot @;
  do i=1 to 5;
    input weight @;
    output;
  end;
drop i;
datalines;
1 17.39 26.93 19.34 22.56 24.49
2 23.63 23.57 23.54 20.56 22.17
3 24.35 24.58 23.79 26.20 21.55
4 25.52 28.02 28.44 25.07 23.39
5 23.25 21.76 29.80 23.09 23.70
6 23.01 22.67 24.70 20.02 26.35
7 23.86 24.19 24.61 26.05 24.18
8 26.00 26.82 28.03 26.27 25.85
9 21.58 22.31 25.03 20.86 26.94
10 22.64 21.05 22.66 29.26 25.02
11 26.38 27.50 23.91 26.80 22.53
12 23.01 23.71 25.26 20.21 22.38
13 23.15 23.53 22.98 21.62 26.99
14 26.83 23.14 24.73 24.57 28.09
15 26.15 26.13 20.57 25.86 24.70
16 25.81 23.22 23.99 23.91 27.57
17 25.53 22.87 25.22 24.30 20.29
18 24.88 24.15 25.29 29.02 24.46
19 22.32 25.96 29.54 25.92 23.44
20 25.63 26.83 20.95 24.80 27.25
21 21.68 21.11 26.07 25.17 27.63
22 26.72 27.05 24.90 30.08 25.22
23 31.58 22.41 23.67 23.47 24.90
24 28.06 23.44 24.92 24.64 27.42
25 21.10 22.34 24.96 26.50 24.51
26 23.80 24.03 24.75 24.82 27.21
27 25.10 26.09 27.21 24.28 22.45
28 25.53 22.79 26.26 25.85 25.64
;

```

A partial listing of DETERGNT is shown in Figure 36.1.

The Data Set DETERGNT	
lot	weight
1	17.39
1	26.93
1	19.34
1	22.56
1	24.49
2	23.63
2	23.57
2	23.54
2	20.56
2	22.17
3	24.35
3	24.58
3	23.79
3	26.20
3	21.55
4	25.52
.	.
.	.
.	.

Figure 36.1. Partial Listing of the Data Set DETERGNT

The data set DETERGNT is said to be in “strung-out” form, since each observation contains the lot number and weight of a single box. The first five observations contain the weights for the first lot, the second five observations contain the weights for the second lot, and so on. Because the variable LOT classifies the observations into rational subgroups, it is referred to as the *subgroup-variable*. The variable WEIGHT contains the weights and is referred to as the *process variable* (or *process* for short).

You can use median and range charts to determine whether the fill process is in control. The following statements create the charts shown in Figure 36.2:

```

title 'Median and Range Charts for Detergent Box Weight';
symbol v=dot;
proc shewhart data=detergnt;
  mrchart weight*lot;
run;

```

This example illustrates the basic form of the MRCHART statement. After the keyword MRCHART, you specify the *process* to analyze (in this case, WEIGHT) followed by an asterisk and the *subgroup-variable* (LOT).

The input data set is specified with the DATA= option in the PROC SHEWHART statement.




Figure 36.2. Median and Range Charts

Each point on the median chart represents the median of the measurements for a particular lot. For instance, the weights for the first lot are 17.39, 19.34, 22.56, 24.49, and 26.93, and consequently, the median plotted for this lot is 22.56. Each point on the range chart represents the range of the measurements for a particular batch. For instance, the range plotted for the first lot is $26.93 - 17.39 = 9.54$. Since all of the points lie within the control limits, you can conclude that the process is in statistical control.

By default, the control limits shown are 3σ limits estimated from the data; the formulas for the limits are given in Table 36.22 on page 1243. You can also read control limits from an input data set; see “Reading Preestablished Control Limits” on page 1229.

For computational details, see “Constructing Charts for Medians and Ranges” on page 1242. For more details on reading raw data, see “DATA= Data Set” on page 1248.

Creating Charts for Medians and Ranges from Summary Data

See SHWMR1
in the SAS/QC
Sample Library

The previous example illustrates how you can create median and range charts using raw data (process measurements). However, in many applications, the data are provided as subgroup summary statistics. This example illustrates how you can use the MRCHART statement with data of this type.

The following data set (DETSUM) provides the data from the preceding example in summarized form. There is exactly one observation for each subgroup (note that the subgroups are still indexed by LOT). The variable WEIGHTM contains the subgroup medians, the variable WEIGHTR contains the subgroup ranges, and the variable WEIGHTN contains the subgroup sample sizes (these are all five).

```

data detsum;
  input lot weightm weightr;
  weightn = 5;
datalines;
  1  22.56  9.54
  2  23.54  3.07
  3  24.35  4.65
  4  25.52  5.05
  5  23.25  8.04
  6  23.01  6.33
  7  24.19  2.19
  8  26.27  2.18
  9  22.31  6.08
10  22.66  8.21
11  26.38  4.97
12  23.01  5.05
13  23.15  5.37
14  24.73  4.95
15  25.86  5.58
16  23.99  4.35
17  24.30  5.24
18  24.88  4.87
19  25.92  7.22
20  25.63  6.30
21  25.17  6.52
22  26.72  5.18
23  23.67  9.17
24  24.92  4.62
25  24.51  5.40
26  24.75  3.41
27  25.10  4.76
28  25.64  3.47
;

```

A partial listing of DETSUM is shown in Figure 36.3.

Summary Data for Detergent Box Weights				
lot	weightm	weightr	weightn	
1	22.56	9.54	5	
2	23.54	3.07	5	
3	24.35	4.65	5	
4	25.52	5.05	5	
5	23.25	8.04	5	
.	.	.	.	
.	.	.	.	
.	.	.	.	

Figure 36.3. The Summary Data Set DETSUM

You can read this data set by specifying it as a HISTORY= data set in the PROC SHEWHART statement, as follows:

Part 9. The CAPABILITY Procedure

```

title 'Median and Range Charts for Weights';
proc shewhart history=detsum lineprinter;
  mrchart weight*lot='*';
run;

```

The charts are shown in Figure 36.4. Since the LINEPRINTER option is included in the PROC SHEWHART statement, line printer output is provided. The asterisk (*) specified in single quotes after the *subgroup-variable* indicates the character used to plot points. This character must follow an equal sign.

Note that WEIGHT is *not* the name of a SAS variable in the data set DETSUM but is, instead, the common prefix for the names of the three SAS variables WEIGHTM, WEIGHTR, and WEIGHTN. The suffix characters *M*, *R*, and *N* indicate *median*, *range*, and *sample size*, respectively. This naming convention enables you to specify three subgroup summary variables in the HISTORY= data set with a single name (WEIGHT), referred to as the *process*. The name LOT specified after the asterisk is the name of the *subgroup-variable*.




Figure 36.4. Median and Range Charts from the Summary Data Set DETSUM

In general, a HISTORY= input data set used with the MRCHART statement must contain the following variables:

- subgroup variable
- subgroup median variable
- subgroup range variable
- subgroup sample size variable

Furthermore, the names of the subgroup median, range, and sample size variables must begin with the prefix *process* specified in the MRCHART statement and end with the special suffix characters *M*, *R*, and *N*, respectively. If the names do not follow this convention, you can use the RENAME option to rename the variables for the duration of the SHEWHART procedure step. Suppose that, instead of the variables WEIGHTM, WEIGHTR, and WEIGHTN, the data set DETSUM contained summary variables named MEDIANS, RANGES, and SIZES. The following statements would temporarily rename MEDIANS, RANGES, and SIZES to WEIGHTM, WEIGHTR, and WEIGHTN, respectively:

```
proc shewhart
  history=detsum (rename=(medians = weightm
                        ranges   = weightr
                        sizes   = weightn ));
  mrchart weight*lot;
run;
```

In summary, the interpretation of *process* depends on the input data set:

- If raw data are read using the DATA= option (as in the previous example), *process* is the name of the SAS variable containing the process measurements.
- If summary data are read using the HISTORY= option (as in this example), *process* is the common prefix for the names containing the summary statistics.

For more information, see “HISTORY= Data Set” on page 1249.

Saving Summary Statistics

In this example, the MRCHART statement is used to create a summary data set that can be read later by the SHEWHART procedure (as in the preceding example). The following statements read measurements from the data set DETERGNT and create a summary data set named DETHIST:

```
title 'Summary Data Set DETHIST for Detergent Box Weights';
proc shewhart data=detergnt;
  mrchart weight*lot / outhistory = dethist
                    nochart;
run;
```

See SHWMR1 in the SAS/QC Sample Library

The OUTHISTORY= option names the output data set, and the NOCHART option suppresses the display of the charts, which would be identical to the charts in Figure 36.2. Options such as OUTHISTORY= and NOCHART are specified after the slash (/) in the MRCHART statement. A complete list of options is presented in the “Syntax” section on page 1231.

Figure 36.5 contains a partial listing of DETHIST.

Summary Data Set DETHIST for Detergent Box Weights				
lot	weightM	weightR	weightN	
1	22.56	9.54	5	
2	23.54	3.07	5	
3	24.35	4.65	5	
4	25.52	5.05	5	
5	23.25	8.04	5	
.	.	.	.	
.	.	.	.	
.	.	.	.	

Figure 36.5. The Summary Data Set DETHIST

There are four variables in the data set DETHIST.

- LOT contains the subgroup index.
- WEIGHTM contains the subgroup medians.
- WEIGHTR contains the subgroup ranges.
- WEIGHTN contains the subgroup sample sizes.

Note that the summary statistic variables are named by adding the suffix characters *M*, *R*, and *N* to the *process* WEIGHT specified in the MRCHART statement. In other words, the variable naming convention for OUTHISTORY= data sets is the same as that for HISTORY= data sets.

For more information, see “OUTHISTORY= Data Set” on page 1245.

Saving Control Limits

See SHWMR1
in the SAS/QC
Sample Library

You can save the control limits for median and range charts in a SAS data set; this enables you to apply the control limits to future data (see “Reading Preestablished Control Limits” on page 1229) or modify the limits with a DATA step program.

The following statements read measurements from the data set DETERGNT (see page 1220) and save the control limits displayed in Figure 36.2 in a data set named DETLIM:

```

title 'Control Limits for Detergent Box Weights';
proc shewhart data=detergnt;
  mrchart weight*lot / outlimits=detlim
                    nochart;
run;

```

The OUTLIMITS= option names the data set containing the control limits, and the NOCHART option suppresses the display of the charts. The data set DETLIM is listed in Figure 36.6.

Control Limits for Detergent Box Weights						
<u>_VAR_</u>	<u>_SUBGRP_</u>	<u>_TYPE_</u>	<u>_LIMITN_</u>	<u>_ALPHA_</u>	<u>_SIGMAS_</u>	<u>_LCLM_</u>
weight	lot	ESTIMATE	5	.002909021	3	20.7554
<u>_MEAN_</u>	<u>_UCLM_</u>	<u>_LCLR_</u>	<u>_R_</u>	<u>_UCLR_</u>	<u>_STDDEV_</u>	
24.4996	28.2439	0	5.42036	11.4613	2.33041	

Figure 36.6. The Data Set DETLIM Containing Control Limit Information

The data set DETLIM contains one observation with the limits for *process* WEIGHT. The variables _LCLM_ and _UCLM_ contain the control limits for the medians, and the variable _MEAN_ contains the central line. The variables _LCLR_ and _UCLR_ contain the control limits for the ranges, and the variable _R_ contains the central line. The values of _MEAN_ and _STDDEV_ are estimates of the process mean and process standard deviation σ . The value of _LIMITN_ is the nominal sample size associated with the control limits, and the value of _SIGMAS_ is the multiple of σ associated with the control limits. The variables _VAR_ and _SUBGRP_ are bookkeeping variables that save the *process* and *subgroup-variable*. The variable _TYPE_ is a bookkeeping variable that indicates whether the values of _MEAN_ and _STDDEV_ are estimates or standard values. For more information, see “OUTLIMITS= Data Set” on page 1244.

You can create an output data set containing both control limits and summary statistics with the OUTTABLE= option, as illustrated by the following statements:

```

title 'Summary Statistics and Control Limit Information';
proc shewhart data=detergnt;
  mrchart weight*lot / outtable=dtable
                    nochart;
run;

```

This data set contains one observation for each subgroup sample. The variables _SUBMED_, _SUBR_, and _SUBN_ contain the subgroup medians, subgroup ranges, and subgroup sample sizes. The variables _LCLM_ and _UCLM_ contain the control limits for the median chart, and the variables _LCLR_ and _UCLR_ contain the control limits for the range chart. The variable _MEAN_ contains the central line for the median chart, and the variable _R_ contains the central line for the range chart. The variables _VAR_ and BATCH contain the *process* name and values of the *subgroup-variable*, respectively. For more information, see “OUTTABLE= Data Set” on page 1246.

The data set DTABLE is listed in Figure 36.7.

Summary Statistics and Control Limit Information												
	S L			S			E			E		
	I I			U			X L			X		
	G M S			L B			M U			L S		
V	M I U			C M			E C			L C U		
A	l A T B			L E			A L			I L B		
R	o S N N			M D			N M			M R R		
t	- - -			- - -			- - -			- - -		
weight	1	3	5	5	20.7554	22.56	24.4996	28.2439	0	9.54	5.42036	11.4613
weight	2	3	5	5	20.7554	23.54	24.4996	28.2439	0	3.07	5.42036	11.4613
weight	3	3	5	5	20.7554	24.35	24.4996	28.2439	0	4.65	5.42036	11.4613
weight	4	3	5	5	20.7554	25.52	24.4996	28.2439	0	5.05	5.42036	11.4613
weight	5	3	5	5	20.7554	23.25	24.4996	28.2439	0	8.04	5.42036	11.4613
weight	6	3	5	5	20.7554	23.01	24.4996	28.2439	0	6.33	5.42036	11.4613
weight	7	3	5	5	20.7554	24.19	24.4996	28.2439	0	2.19	5.42036	11.4613
weight	8	3	5	5	20.7554	26.27	24.4996	28.2439	0	2.18	5.42036	11.4613
weight	9	3	5	5	20.7554	22.31	24.4996	28.2439	0	6.08	5.42036	11.4613
weight	10	3	5	5	20.7554	22.66	24.4996	28.2439	0	8.21	5.42036	11.4613
weight	11	3	5	5	20.7554	26.38	24.4996	28.2439	0	4.97	5.42036	11.4613
weight	12	3	5	5	20.7554	23.01	24.4996	28.2439	0	5.05	5.42036	11.4613
weight	13	3	5	5	20.7554	23.15	24.4996	28.2439	0	5.37	5.42036	11.4613
weight	14	3	5	5	20.7554	24.73	24.4996	28.2439	0	4.95	5.42036	11.4613
weight	15	3	5	5	20.7554	25.86	24.4996	28.2439	0	5.58	5.42036	11.4613
weight	16	3	5	5	20.7554	23.99	24.4996	28.2439	0	4.35	5.42036	11.4613
weight	17	3	5	5	20.7554	24.30	24.4996	28.2439	0	5.24	5.42036	11.4613
weight	18	3	5	5	20.7554	24.88	24.4996	28.2439	0	4.87	5.42036	11.4613
weight	19	3	5	5	20.7554	25.92	24.4996	28.2439	0	7.22	5.42036	11.4613
weight	20	3	5	5	20.7554	25.63	24.4996	28.2439	0	6.30	5.42036	11.4613
weight	21	3	5	5	20.7554	25.17	24.4996	28.2439	0	6.52	5.42036	11.4613
weight	22	3	5	5	20.7554	26.72	24.4996	28.2439	0	5.18	5.42036	11.4613
weight	23	3	5	5	20.7554	23.67	24.4996	28.2439	0	9.17	5.42036	11.4613
weight	24	3	5	5	20.7554	24.92	24.4996	28.2439	0	4.62	5.42036	11.4613
weight	25	3	5	5	20.7554	24.51	24.4996	28.2439	0	5.40	5.42036	11.4613
weight	26	3	5	5	20.7554	24.75	24.4996	28.2439	0	3.41	5.42036	11.4613
weight	27	3	5	5	20.7554	25.10	24.4996	28.2439	0	4.76	5.42036	11.4613
weight	28	3	5	5	20.7554	25.64	24.4996	28.2439	0	3.47	5.42036	11.4613

Figure 36.7. The Data Set DTABLE

An OUTTABLE= data set can be read later as a TABLE= data set. For example, the following statements read DTABLE and display charts (not shown here) identical to those in Figure 36.2:

```

title 'Median and Range Charts for Detergent Box Weight';
proc shewhart table=dtable;
    mrchart weight*lot;
run;

```

Because the SHEWHART procedure simply displays the information in a TABLE= data set, you can use TABLE= data sets to create specialized control charts (see Chapter 49, “Specialized Control Charts”). For more information, see “TABLE= Data Set” on page 1250.

Reading Prestablished Control Limits

In the previous example, the OUTLIMITS= data set DETLIM saved control limits computed from the measurements in DETERGNT. This example shows how these limits can be applied to new data provided in the following data set:

See SHWMR1
in the SAS/QC
Sample Library

```

data detergt2;
  input lot @;
  do i=1 to 5;
    input weight @;
    output;
  end;
  drop i;
  datalines;
29 16.66 27.49 18.87 22.53 24.72
30 23.74 23.67 23.64 20.26 22.09
31 24.56 24.82 23.92 26.67 21.38
32 25.89 28.73 29.21 25.38 23.47
33 23.32 21.61 30.75 23.13 23.82
34 23.04 22.65 24.96 19.64 26.84
35 24.01 24.38 24.86 26.50 24.37
36 26.43 27.36 28.74 26.74 26.27
37 21.41 22.24 25.34 20.59 27.51
38 22.62 20.81 22.64 30.15 25.32
39 26.86 28.14 24.06 27.35 22.49
40 23.03 23.83 25.59 19.85 22.33
41 23.19 23.63 23.00 21.46 27.57
42 27.38 23.18 24.99 24.81 28.82
43 26.60 26.58 20.26 26.27 24.96
44 26.22 23.28 24.15 24.06 28.23
45 25.90 22.88 25.55 24.50 19.95
46 16.66 27.49 18.87 22.53 24.72
47 23.74 23.67 23.64 20.26 22.09
48 24.56 24.82 23.92 26.67 21.38
49 25.89 28.73 29.21 25.38 23.47
50 23.32 21.61 30.75 23.13 23.82
;

```

The following statements create median and range charts for the data in DETERGT2 using the control limits in DETLIM:

```

title 'Median and Range Charts for Weights';
symbol v=dot;
proc shewhart data=detergt2 limits=detlim;
  mrchart weight*lot;
run;

```

Part 9. The CAPABILITY Procedure

The charts are shown in Figure 36.8. The LIMITS= option in the PROC SHEWHART statement specifies the data set containing the control limits. By default,* this information is read from the first observation in the LIMITS= data set for which

- the value of _VAR_ matches the *process* name WEIGHT
- the value of _SUBGRP_ matches the *subgroup-variable* name LOT




Figure 36.8. Median and Range Charts for Second Set of Detergent Box Weights
The charts indicate that the process is in control, since all the medians and ranges lie within the control limits.

In this example, the LIMITS= data set was created in a previous run of the SHEWHART procedure. You can also create a LIMITS= data set with the DATA step. See “LIMITS= Data Set” on page 1248 for details concerning the variables that you must provide.

*In Release 6.09 and in earlier releases, it is also necessary to specify the READLIMITS option to read control limits from a LIMITS= data set.

Syntax

The basic syntax for the MRCHART statement is as follows:

```
MRCHART process*subgroup-variable ;
```

The general form of this syntax is as follows:

```
MRCHART (processes)*subgroup-variable <( block-variables ) >  
      < =symbol-variable | ='character' > < / options >;
```

You can use any number of MRCHART statements in the SHEWHART procedure. The components of the MRCHART statement are described as follows.

process

processes

identify one or more processes to be analyzed. The specification of *process* depends on the input data set specified in the PROC SHEWHART statement.

- If raw data are read from a DATA= data set, *process* must be the name of the variable containing the raw measurements. For an example, see “Creating Charts for Medians and Ranges from Raw Data” on page 1220.
- If summary data are read from a HISTORY= data set, *process* must be the common prefix of the summary variables in the HISTORY= data set. For an example, see “Creating Charts for Medians and Ranges from Summary Data” on page 1222.
- If summary data and control limits are read from a TABLE= data set, *process* must be the value of the variable `_VAR_` in the TABLE= data set. For an example, see “Saving Control Limits” on page 1226.

A *process* is required. If you specify more than one *process*, enclose the list in parentheses. For example, the following statements request distinct median and range charts for WEIGHT, LENGTH, and WIDTH:

```
proc shewhart data=measures;  
      mrchart (weight length width)*day;  
run;
```

subgroup-variable

is the variable that identifies subgroups in the data. The *subgroup-variable* is required. In the preceding MRCHART statement, DAY is the subgroup variable. For details, see “Subgroup Variables” on page 1534.

block-variables

are optional variables that group the data into blocks of consecutive subgroups. The blocks are labeled in a legend, and each *block-variable* provides one level of labels in the legend. See “Displaying Stratification in Blocks of Observations” on page 1684 for an example.

symbol-variable

is an optional variable whose levels (unique values) determine the symbol marker or character used to plot the medians and ranges.

- If you produce a chart on a line printer, an ‘A’ is displayed for the points corresponding to the first level of the *symbol-variable*, a ‘B’ is displayed for the points corresponding to the second level, and so on.
- If you produce a chart on a graphics device, distinct symbol markers are displayed for points corresponding to the various levels of the *symbol-variable*. You can specify the symbol markers with SYMBOLn statements. See “Displaying Stratification in Levels of a Classification Variable” on page 1683 for an example.

character

specifies a plotting character for charts produced on line printers. For example, the following statements create median and range charts using an asterisk (*) to plot the points:

```
proc shewhart data=values;
    mrchart weight*day='*';
run;
```

options

enhance the appearance of the charts, request additional analyses, save results in data sets, and so on. The “Summary of Options” section, which follows, lists all options by function. Chapter 46, “Dictionary of Options,” describes each option in detail.

Summary of Options

The following tables list the MRCHART statement options by function. For complete descriptions, see Chapter 46, “Dictionary of Options.”

Table 36.1. Tabulation Options

TABLE	creates a basic table of subgroup medians, subgroup ranges, subgroup sample sizes, and control limits
TABLEALL	is equivalent to the options TABLE, TABLECENTRAL, TABLEID, TABLELEGEND, TABLEOUT, and TABLETESTS
TABLECENTRAL	augments basic table with values of central lines
TABLEID	augments basic table with columns for ID variables
TABLELEGEND	augments basic table with legend for tests for special causes
TABLEOUTLIM	augments basic table with columns indicating control limits exceeded
TABLETESTS	augments basic table with a column indicating which tests for special causes are positive

Note that specifying (EXCEPTIONS) after a tabulation option creates a table for exceptional points only.

Table 36.2. Options for Specifying Tests for Special Causes

NO3SIGMACHECK	allows tests to be applied with control limits other than 3σ limits
TESTS= <i>value-list</i> <i>customized-pattern-list</i>	specifies tests for special causes for the median chart
TESTS2= <i>value-list</i> <i>customized-pattern-list</i>	specifies tests for special causes for the range chart
TEST2RUN= <i>n</i>	specifies length of pattern for Test 2
TEST3RUN= <i>n</i>	specifies length of pattern for Test 3
TESTACROSS	applies tests across <i>phase</i> boundaries
TESTLABEL= <i>'label'</i> <i>(variable)</i> <i>keyword</i>	provides labels for points where test is positive
TESTLABEL <i>n</i> = <i>'label'</i>	specifies label for <i>n</i> th test for special causes
TESTNMETHOD= STANDARDIZE	applies tests to standardized chart statistics
TESTOVERLAP	performs tests on overlapping patterns of points
ZONELABELS	adds labels A, B, and C to zone lines for median chart
ZONE2LABELS	adds labels A, B, and C to zone lines for range chart
ZONES	adds lines to median chart delineating zones A, B, and C
ZONES2	adds lines to range chart delineating zones A, B, and C
ZONEVALPOS= <i>n</i>	specifies position of ZONEVALUES and ZONE2VALUES labels
ZONEVALUES	labels median chart zone lines with their values
ZONE2VALUES	labels range chart zone lines with their values

Table 36.3. Graphical Options for Displaying Tests for Special Causes

CTESTS= <i>color</i> <i>test-color-list</i>	specifies color for labels used to identify points where test is positive
CZONES= <i>color</i>	specifies color for lines and labels delineating zones A, B, and C
LABELFONT= <i>font</i>	specifies software font for labels at points where test is positive (alias for the TESTFONT= option)
LABELHEIGHT= <i>value</i>	specifies height of labels at points where test is positive (alias for the TESTHEIGHT= option)
LTESTS= <i>linetype</i>	specifies type of line connecting points where test is positive
LZONES= <i>linetype</i>	specifies line type for lines delineating zones A, B, and C
TESTFONT= <i>font</i>	specifies software font for labels at points where test is positive
TESTHEIGHT= <i>value</i>	specifies height of labels at points where test is positive

Table 36.4. Line Printer Options for Displaying Tests for Special Causes

TESTCHAR= <i>'character'</i>	specifies character for line segments that connect any sequence of points for which a test for special causes is positive
ZONECHAR= <i>'character'</i>	specifies character for lines that delineate zones for tests for special causes

Table 36.5. Reference Line Options

CHREF= <i>color</i>	specifies color for lines requested by HREF= and HREF2= options
CVREF= <i>color</i>	specifies color for lines requested by VREF= and VREF2= options
HREF= <i>values</i> <i>SAS-data-set</i>	specifies position of reference lines perpendicular to horizontal axis on median chart
HREF2= <i>values</i> <i>SAS-data-set</i>	specifies position of reference lines perpendicular to horizontal axis on range chart
HREFCHAR= <i>'character'</i>	specifies line character for HREF= and HREF2= lines
HREFDATA= <i>SAS-data-set</i>	specifies position of reference lines perpendicular to horizontal axis on median chart
HREF2DATA= <i>SAS-data-set</i>	specifies position of reference lines perpendicular to horizontal axis on range chart
HREFLABELS= <i>'label1'...'labeln'</i>	specifies labels for HREF= lines
HREF2LABELS= <i>'label1'...'labeln'</i>	specifies labels for HREF2= lines
HREFLABPOS= <i>n</i>	specifies position of HREFLABELS= and HREF2LABELS= labels
LHREF= <i>linetype</i>	specifies line type for HREF= and HREF2= lines
LVREF= <i>linetype</i>	specifies line type for VREF= and VREF2= lines
NOBYREF	specifies that reference line information in a data set is to be applied uniformly to charts created for all BY groups
VREF= <i>values</i> <i>SAS-data-set</i>	specifies position of reference lines perpendicular to vertical axis on median chart
VREF2= <i>values</i> <i>SAS-data-set</i>	specifies position of reference lines perpendicular to vertical axis on range chart
VREFCHAR= <i>'character'</i>	specifies line character for VREF= and VREF2= lines
VREFLABELS= <i>'label1'...'labeln'</i>	specifies labels for VREF= lines
VREF2LABELS= <i>'label1'...'labeln'</i>	specifies labels for VREF2= lines
VREFLABPOS= <i>n</i>	specifies position of VREFLABELS= and VREF2LABELS= labels

Table 36.6. Clipping Options

CCLIP= <i>color</i>	specifies color for plot symbol for clipped points
CLIPCHAR= <i>'character'</i>	specifies plot character for clipped points
CLIPFACTOR= <i>value</i>	determines extent to which extreme points are clipped
CLIPLEGEND= <i>'string'</i>	specifies text for clipping legend
CLIPLEGPOS= <i>keyword</i>	specifies position of clipping legend
CLIPSUBCHAR= <i>'character'</i>	specifies substitution character for CLIPLEGEND= text
CLIPSYMBOL= <i>symbol</i>	specifies plot symbol for clipped points
CLIPSYMBOLHT= <i>value</i>	specifies symbol marker height for clipped points

Table 36.7. Axis and Axis Label Options

CAXIS= <i>color</i>	specifies color for axis lines and tick marks
CFRAME= <i>color</i> <i>(color-list)</i>	specifies fill colors for frame for plot area
CTEXT= <i>color</i>	specifies color for tick mark values and axis labels
HAXIS= <i>values</i> <i>AXISn</i>	specifies major tick mark values for horizontal axis
HEIGHT= <i>value</i>	specifies height of axis label and axis legend text
HMINOR= <i>n</i>	specifies number of minor tick marks between major tick marks on horizontal axis
HOFFSET= <i>value</i>	specifies length of offset at both ends of horizontal axis
INTSTART= <i>value</i>	specifies first major tick mark value for numeric horizontal axis
NOHLABEL	suppresses label for horizontal axis
NOTICKREP	specifies that only the first occurrence of repeated, adjacent subgroup values is to be labeled on horizontal axis
NOTRUNC	suppresses vertical axis truncation at zero applied by default to range chart
NOVANGLE	requests vertical axis labels that are strung out vertically
SKIPLABELS= <i>n</i>	specifies thinning factor for tick mark labels on horizontal axis
SPLIT= <i>'character'</i>	specifies splitting character for axis labels
TURNHLABELS	requests horizontal axis labels that are strung out vertically
VAXIS= <i>values</i> <i>AXISn</i>	specifies major tick mark values for vertical axis of median chart
VAXIS2= <i>values</i> <i>AXISn</i>	specifies major tick mark values for vertical axis of range chart
VMINOR= <i>n</i>	specifies number of minor tick marks between major tick marks on vertical axis
VOFFSET= <i>value</i>	specifies length of offset at both ends of vertical axis
VZERO	forces origin to be included in vertical axis for primary chart
VZERO2	forces origin to be included in vertical axis for secondary chart
WAXIS= <i>n</i>	specifies width of axis lines

Table 36.8. Options for Plotting and Labeling Points

ALLLABEL=VALUE (variable)	labels every point on median chart
ALLLABEL2=VALUE (variable)	labels every point on range chart
CCONNECT=color	specifies color for line segments that connect points on chart
CFRAMELAB=color	specifies fill color for frame around labeled points
CNEEDLES=color	specifies color for needles that connect points to central line
CONNECTCHAR= 'character'	specifies character used to form line segments that connect points on chart
COUT=color	specifies color for portions of line segments that connect points outside control limits
COUTFILL=color	specifies color for shading areas between the connected points and control limits outside the limits
NEEDLES	connects points to central line with vertical needles
NOCONNECT	suppresses line segments that connect points on chart
OUTLABEL=VALUE (variable)	labels points outside control limits on median chart
OUTLABEL2=VALUE (variable)	labels points outside control limits on range chart
SYMBOLCHARS= 'characters'	specifies characters indicating <i>symbol-variable</i>
SYMBOLLEGEND= NONE name	specifies LEGEND statement for levels of <i>symbol-variable</i>
SYMBOLORDER= keyword	specifies order in which symbols are assigned for levels of <i>symbol-variable</i>
TURNALL TURNOUT	turns point labels so that they are strung out vertically

Table 36.9. Options for Specifying Control Limits

ALPHA=value	requests probability limits for control charts
LIMITN=n VARYING	specifies either nominal sample size for fixed control limits or varying control limits
NOREADLIMITS	computes control limits for each <i>process</i> from the data rather than from a LIMITS= data set (Release 6.10 and later releases)
READALPHA	reads _ALPHA_ instead of _SIGMAS_ from a LIMITS= data set
READINDEXES=ALL 'label1'...'labeln'	reads multiple sets of control limits for each <i>process</i> from a LIMITS= data set
READLIMITS	reads single set of control limits for each <i>process</i> from a LIMITS= data set (Release 6.09 and earlier releases)
SIGMAS=k	specifies width of control limits in terms of multiple <i>k</i> of standard error of plotted statistic

Table 36.10. Options for Displaying Control Limits

CINFILL= <i>color</i>	specifies color for area inside control limits
CLIMITS= <i>color</i>	specifies color of control limits, central line, and related labels
LCLLABEL= <i>'label'</i>	specifies label for lower control limit on median chart
LCLLABEL2= <i>'label'</i>	specifies label for lower control limit on range chart
LIMLABSUBCHAR= <i>'character'</i>	specifies a substitution character for labels provided as quoted strings; the character is replaced with the value of the control limit
LLIMITS= <i>linetype</i>	specifies line type for control limits
NDECIMAL= <i>n</i>	specifies number of digits to right of decimal place in default labels for control limits and central line on median chart
NDECIMAL2= <i>n</i>	specifies number of digits to right of decimal place in default labels for control limits and central line on range chart
NOCTL	suppresses display of central line on median chart
NOCTL2	suppresses display of central line on range chart
NOLCL	suppresses display of lower control limit on median chart
NOLCL2	suppresses display of lower control limit on range chart
NOLIMITLABEL	suppresses labels for control limits and central lines
NOLIMITS	suppresses display of control limits
NOLIMITSFRAME	suppresses default frame around control limit information when multiple sets of control limits are read from LIMITS= data set
NOLIMITSLEGEND	suppresses legend for control limits
NOLIMIT0	suppresses display of zero lower control limit on range chart
NOUCL	suppresses display of upper control limit on median chart
NOUCL2	suppresses display of upper control limit on range chart
RSYMBOL= <i>'string'</i> <i>keyword</i>	specifies label for central line on range chart
UCLLABEL= <i>'string'</i>	specifies label for upper control limit on median chart
UCLLABEL2= <i>'string'</i>	specifies label for upper control limit on range chart
WLIMITS= <i>n</i>	specifies width for control limits and central line
XSYMBOL= <i>'string'</i> <i>keyword</i>	specifies label for central line on median chart

Table 36.11. Grid Options

ENDGRID	adds grid after last plotted point
GRID	adds grid to control chart
LENDGRID= <i>linetype</i>	specifies line type for grid requested with the ENDGRID option
LGRID= <i>linetype</i>	specifies line type for grid requested with the GRID option
WGRID= <i>n</i>	specifies width of grid lines

Table 36.12. Phase Options

CPHASEBOX= <i>color</i>	specifies color for box enclosing all plotted points for a phase
CPHASEBOX- CONNECT= <i>color</i>	specifies color for line segments connecting adjacent enclosing boxes
CPHASEBOXFILL= <i>color</i>	specifies fill color for box enclosing all plotted points for a phase
CPHASELEG= <i>color</i>	specifies text color for <i>phase</i> legend
CPHASEMEAN- CONNECT= <i>color</i>	specifies color for line segments connecting average value points within a phase
NOPHASEFRAME	suppresses default frame for <i>phase</i> legend
OUTPHASE= <i>'string'</i>	specifies value of <code>_PHASE_</code> in the OUTHISTORY= data set
PHASEBREAK	disconnects last point in a <i>phase</i> from first point in next <i>phase</i>
PHASELABTYPE= <i>value</i> <i>keyword</i>	specifies text size of <i>phase</i> legend
PHASELEGEND	displays <i>phase</i> labels in a legend across top of chart
PHASELIMITS	labels control limits for each phase, provided they are constant within that phase
PHASEMEANSYMBOL= <i>symbol</i>	specifies symbol marker for average of values within a phase
PHASEREF	delineates <i>phases</i> with vertical reference lines
READPHASES= ALL <i>'label1' ...'labeln'</i>	specifies <i>phases</i> to be read from an input data set

Table 36.13. Specification Limit Options

CIINDICES=(ALPHA= <i>value</i> TYPE= <i>keyword</i>)	specifies α value and type for computing capability index confidence limits
LSL= <i>value-list</i>	specifies list of lower specification limits
TARGET= <i>value-list</i>	specifies list of target values
USL= <i>value-list</i>	specifies list of upper specification limits

Table 36.14. Process Mean and Standard Deviation Options

MEDCENTRAL= <i>keyword</i>	specifies method for estimating process mean μ
MU0= <i>value</i>	specifies known value μ_0 for process mean μ
SIGMA0= <i>value</i>	specifies known value σ_0 for process standard deviation σ
SMETHOD= <i>keyword</i>	specifies method for estimating process standard deviation σ
TYPE= <i>keyword</i>	identifies whether parameters are estimates or standard values and specifies value of <code>_TYPE_</code> in the OUTLIMITS= data set

Table 36.15. Input Data Set Options

MISSBREAK	specifies that observations with missing values are not to be processed
-----------	---

Table 36.16. Output Data Set Options

OUTHISTORY= <i>SAS-data-set</i>	creates output data set containing subgroup summary statistics
OUTINDEX='string'	specifies value of <code>_INDEX_</code> in the OUTLIMITS= data set
OUTLIMITS= <i>SAS-data-set</i>	creates output data set containing control limits
OUTTABLE= <i>SAS-data-set</i>	creates output data set containing subgroup summary statistics and control limits

Table 36.17. Plot Layout Options

ALLN	plots summary statistics for all subgroups
BILEVEL	creates control charts using half-screens and half-pages
EXCHART	creates control charts for a process variable only when exceptions occur
INTERVAL= <i>keyword</i>	specifies natural time interval between consecutive subgroup positions when time, date, or datetime format is used
MAXPANELS= <i>n</i>	specifies maximum number of pages or screens for chart
NMARKERS	requests special markers for points corresponding to sample sizes not equal to nominal sample size for fixed control limits
NOCHART	suppresses creation of charts
NOCHART2	suppresses creation of range chart
NOFRAME	suppresses frame for plot area
NOLEGEND	suppresses legend for subgroup sample sizes
NPANELPOS= <i>n</i>	specifies number of subgroup positions per panel on each chart
REPEAT	repeats last subgroup position on panel as first subgroup position of next panel
SEPARATE	displays median and range charts on separate screens or pages
TOTPANELS= <i>n</i>	specifies number of pages or screens to be used to display chart
YPCT1= <i>value</i>	specifies length of vertical axis on median chart as a percentage of sum of lengths of vertical axes for median and range charts
ZEROSTD	displays median and range chart regardless of whether $\hat{\sigma} = 0$

Table 36.18. Graphical Enhancement Options

ANNOTATE= <i>SAS-data-set</i>	specifies annotate data set that adds features to median chart
ANNOTATE2= <i>SAS-data-set</i>	specifies annotate data set that adds features to range chart
DESCRIPTION= <i>'string'</i>	specifies string that appears in the description field of the PROC GREPLAY master menu for median chart
DESCRIPTION2= <i>'string'</i>	specifies string that appears in the description field of the PROC GREPLAY master menu for range chart
FONT= <i>font</i>	specifies software font for labels and legends on charts
NAME= <i>'string'</i>	specifies name that appears in the name field of the PROC GREPLAY master menu for median chart
NAME2= <i>'string'</i>	specifies name that appears in the name field of the PROC GREPLAY master menu for range chart
PAGENUM= <i>'string'</i>	specifies the form of the label used in pagination
PAGENUMPOS= <i>keyword</i>	specifies the position of the page number requested with the PAGENUM= option

Table 36.19. Star Options

CSTARCIRCLES= <i>color</i>	specifies color for STARCIRCLES= circles
CSTARFILL= <i>color</i> <i>(variable)</i>	specifies color for filling stars
CSTAROUT= <i>color</i>	specifies outline color for stars exceeding inner or outer circles
CSTARS= <i>color</i> <i>(variable)</i>	specifies color for outlines of stars
LSTARCIRCLES= <i>linetypes</i>	specifies line types for STARCIRCLES= circles
LSTARS= <i>linetype</i> <i>(variable)</i>	specifies line types for outlines of STARVERTICES= stars
STARBDRADIUS= <i>value</i>	specifies radius of outer bound circle for vertices of stars
STARCIRCLES= <i>value-list</i>	specifies reference circles for stars
STARINRADIUS= <i>value</i>	specifies inner radius of stars
STARLABEL= <i>keyword</i>	specifies vertices to be labeled
STARLEGEND= <i>keyword</i>	specifies style of legend for star vertices
STARLEGENDLAB= <i>'label'</i>	specifies label for STARLEGEND= legend
STAROUTRADIUS= <i>value</i>	specifies outer radius of stars
STARSPEC= <i>value</i> <i>SAS-data-set</i>	specifies method used to standardize vertex variables
STARSTART= <i>value</i>	specifies angle for first vertex
STARTYPE= <i>keyword</i>	specifies graphical style of star
STARVERTICES= <i>variable</i> <i>(variables)</i>	superimposes star at each point on median chart
WSTARCIRCLES= <i>n</i>	specifies width of STARCIRCLES= circles
WSTARS= <i>n</i>	specifies width of STARVERTICES= stars

Table 36.20. Block Variable Legend Options

BLOCKLABELPOS= <i>keyword</i>	specifies position of label for <i>block-variable</i> legend
BLOCKLABTYPE= <i>value keyword</i>	specifies text size of <i>block-variable</i> legend
BLOCKPOS= <i>n</i>	specifies vertical position of <i>block-variable</i> legend
BLOCKREP	repeats identical consecutive labels in <i>block-variable</i> legend
CBLOCKLAB= <i>color</i>	specifies color for filling background in <i>block-variable</i> legend
CBLOCKVAR= <i>variable </i> <i>(variables)</i>	specifies one or more variables whose values are colors for filling background of <i>block-variable</i> legend

Table 36.21. Options for Interactive Control Charts

HTML=(<i>variable</i>)	specifies a variable whose values are URLs to be associated with subgroups
HTML_LEGEND= <i>(variable)</i>	specifies a variable whose values are URLs to be associated with symbols in the symbol legend
TESTURLS= <i>SAS-data-set</i>	associates URLs with tests for special causes
WEBOUT= <i>SAS-data-set</i>	creates an OUTTABLE= data set with additional graphics coordinate data

Details

Constructing Charts for Medians and Ranges

The following notation is used in this section:

μ	process mean (expected value of the population of measurements)
σ	process standard deviation (standard deviation of the population of measurements)
\bar{X}_i	mean of measurements in i^{th} subgroup
R_i	range of measurements in i^{th} subgroup
n_i	sample size of i^{th} subgroup
N	the number of subgroups
x_{ij}	j^{th} measurement in the i^{th} subgroup, $j = 1, 2, 3, \dots, n_i$
$x_{i(j)}$	j^{th} largest measurement in the i^{th} subgroup. Then
	$x_{i(1)} \leq x_{i(2)} \leq \dots \leq x_{i(n_i)}$
$\bar{\bar{X}}$	weighted average of subgroup means
M_i	median of the measurements in the i^{th} subgroup:
	$M_i = \begin{cases} x_{i((n_i+1)/2)} & \text{if } n_i \text{ is odd} \\ (x_{i(n_i/2)} + x_{i((n_i/2)+1)})/2 & \text{if } n_i \text{ is even} \end{cases}$
\bar{M}	average of the subgroup medians:
	$\bar{M} = (n_1 M_1 + \dots + n_N M_N) / (n_1 + \dots + n_N)$
\tilde{M}	median of the subgroup medians. Denote the j^{th} largest median by $M_{(j)}$ so that $M_{(1)} \leq M_{(2)} \leq \dots \leq M_{(N)}$.
	$\tilde{M} = \begin{cases} M_{((N+1)/2)} & \text{if } N \text{ is odd} \\ (M_{(N/2)} + M_{(N/2)+1})/2 & \text{if } N \text{ is even} \end{cases}$
$e_M(n)$	standard error of the median of n independent, normally distributed variables with unit standard deviation (the value of $e_M(n)$ can be calculated with the STD MED function in a DATA step)
$Q_p(n)$	$100p^{\text{th}}$ percentile ($0 < p < 1$) of the distribution of the median of n independent observations from a normal population with unit standard deviation
$d_2(n)$	expected value of the range of n independent normally distributed variables with unit standard deviation
$d_3(n)$	standard error of the range of n independent observations from a normal population with unit standard deviation
z_p	$100p^{\text{th}}$ percentile of the standard normal distribution
$D_p(n)$	$100p^{\text{th}}$ percentile of the distribution of the range of n independent observations from a normal population with unit standard deviation

Plotted Points

Each point on a median chart indicates the value of a subgroup median (M_i). For example, if the tenth subgroup contains the values 12, 15, 19, 16, and 14, the value plotted for this subgroup is $M_{10} = 15$. Each point on a range chart indicates the value of a subgroup range (R_i). For example, the value plotted for the tenth subgroup is $R_{10} = 19 - 12 = 7$.

Central Lines

On a median chart, the value of the central line indicates an estimate for μ , which is computed as

- \bar{M} by default
- $\overline{\bar{X}}$ when you specify MEDCENTRAL=AVGMEAN
- \tilde{M} when you specify MEDCENTRAL=MEDMED
- μ_0 when you specify μ_0 with the MU0= option

On the range chart, by default, the central line for the i^{th} subgroup indicates an estimate for the expected value of R_i , which is computed as $d_2(n_i)\hat{\sigma}$, where $\hat{\sigma}$ is an estimate of σ . If you specify a known value (σ_0) for σ , the central line indicates the value of $d_2(n_i)\sigma_0$. The central line on the range chart varies with n_i .

Control Limits

You can compute the limits

- as a specified multiple (k) of the standard errors of M_i and R_i above and below the central line. The default limits are computed with $k = 3$ (these are referred to as 3σ limits).
- as probability limits defined in terms of α , a specified probability that M_i or R_i exceeds its limits

The following table provides the formulas for the limits:

Table 36.22. Limits for Median and Range Charts

Control Limits	
Median Chart	LCL = lower limit = $\bar{M} - k\hat{\sigma}e_M(n_i)$ UCL = upper limit = $\bar{M} + k\hat{\sigma}e_M(n_i)$
Range Chart	LCL = lower control limit = $\max(d_2(n_i)\hat{\sigma} - kd_3(n_i)\hat{\sigma}, 0)$ UCL = upper control limit = $d_2(n_i)\hat{\sigma} + kd_3(n_i)\hat{\sigma}$
Probability Limits	
Median Chart	LCL = lower limit = $\bar{M} - Q_{\alpha/2}(n_i)\hat{\sigma}$ UCL = upper limit = $\bar{M} + Q_{1-\alpha/2}(n_i)\hat{\sigma}$
Range Chart	LCL = lower limit = $D_{\alpha/2}\hat{\sigma}$ UCL = upper limit = $D_{1-\alpha/2}\hat{\sigma}$

In Table 36.22, replace \bar{M} with $\overline{\bar{X}}$ if you specify MEDCENTRAL=AVGMEAN, and replace \bar{M} with \tilde{M} if you specify MEDCENTRAL=MEDMED. Replace \bar{M} with μ_0 if you specify μ_0 with the MU0= option, and replace $\hat{\sigma}$ with σ_0 if you specify σ_0 with the SIGMA0= option.

The formulas assume that the data are normally distributed. Note that the limits for both charts vary with n_i and that the probability limits for R_i are asymmetric around the central line.

You can specify parameters for the limits as follows:

- Specify k with the SIGMAS= option or with the variable `_SIGMAS_` in a LIMITS= data set.
- Specify α with the ALPHA= option or with the variable `_ALPHA_` in a LIMITS= data set.
- Specify a constant nominal sample size $n_i \equiv n$ for the control limits with the LIMITN= option or with the variable `_LIMITN_` in a LIMITS= data set.
- Specify μ_0 with the MU0= option or with the variable `_MEAN_` in the LIMITS= data set.
- Specify σ_0 with the SIGMA0= option or with the variable `_STDDEV_` in the LIMITS= data set.

Output Data Sets

OUTLIMITS= Data Set

The OUTLIMITS= data set saves control limits and control limit parameters. The following variables can be saved:

Table 36.23. OUTLIMITS= Data Set

Variable	Description
<code>_ALPHA_</code>	probability (α) of exceeding limits
<code>_CP_</code>	capability index C_p
<code>_CPK_</code>	capability index C_{pk}
<code>_CPL_</code>	capability index C_{PL}
<code>_CPM_</code>	capability index C_{pm}
<code>_CPU_</code>	capability index C_{PU}
<code>_INDEX_</code>	optional identifier for the control limits specified with the OUTINDEX= option
<code>_LCLM_</code>	lower control limit for subgroup median
<code>_LCLR_</code>	lower control limit for subgroup range
<code>_LIMITN_</code>	sample size associated with the control limits
<code>_LSL_</code>	lower specification limit
<code>_MEAN_</code>	estimate of process mean (\bar{M} , \tilde{M} , $\bar{\bar{X}}$, or μ_0)
<code>_R_</code>	value of central line on range chart
<code>_SIGMAS_</code>	multiple (k) of standard error of M_i or R_i
<code>_STDDEV_</code>	process standard deviation ($\hat{\sigma}$ or σ_0)
<code>_SUBGRP_</code>	<i>subgroup-variable</i> specified in the MRCHART statement
<code>_TARGET_</code>	target value
<code>_TYPE_</code>	type (estimate or standard value) of <code>_MEAN_</code> and <code>_STDDEV_</code>

Table 36.23. (continued)

Variable	Description
UCLM	upper control limit for subgroup median
UCLR	upper control limit for subgroup range
USL	upper specification limit
VAR	<i>process</i> specified in the XRCHART statement

Notes:

1. If the control limits vary with subgroup sample size, the special missing value V is assigned to the variables _LIMITN_, _LCLM_, _UCLM_, _LCLR_, _R_, and _UCLR_.
2. If the limits are defined in terms of a multiple k of the standard errors of M_i and R_i , the value of _ALPHA_ is computed as $\alpha = 2(1 - F_{med}(k, n))$, where $F_{med}(\cdot, n)$ is the cumulative distribution function of the median of a random sample of n standard normally distributed observations, and n is the value of _LIMITN_. If _LIMITN_ has the special missing value V , this value is assigned to _ALPHA_.
3. If the limits are probability limits, the value of _SIGMAS_ is computed as $k = F_{med}^{-1}(1 - \alpha/2, n)$, where $F_{med}^{-1}(\cdot, n)$ is the inverse distribution function of the median of a random sample of n standard normally distributed observations, and n is the value of _LIMITN_. If _LIMITN_ has the special missing value V , this value is assigned to _SIGMAS_.
4. The variables _CP_, _CPK_, _CPL_, _CPU_, _LSL_, and _USL_ are included only if you provide specification limits with the LSL= and USL= options. The variables _CPM_ and _TARGET_ are included if, in addition, you provide a target value with the TARGET= option. See “Capability Indices” on page 1537 for computational details.
5. Optional BY variables are saved in the OUTLIMITS= data set.

The OUTLIMITS= data set contains one observation for each *process* specified in the MRCHART statement. For an example of an OUTLIMITS= data set, see “Saving Control Limits” on page 1226.

OUTHISTORY= Data Set

The OUTHISTORY= option saves subgroup summary statistics. The following variables are saved:

- the *subgroup-variable*
- a subgroup median variable named by *process* suffixed with M
- a subgroup range variable named by *process* suffixed with R
- a subgroup sample size variable named by *process* suffixed with N

Part 9. The CAPABILITY Procedure

Given a *process* name that contains eight characters, the procedure first shortens the name to its first four characters and its last three characters, and then it adds the suffix. For example, the procedure shortens the *process* DIAMETER to DIAMTER before adding the suffix.

Variables containing subgroup medians, ranges, and sample sizes are created for each *process* specified in the MRCHART statement. For example, consider the following statements:

```
proc shewhart data=steel;
    mrchart (width diameter)*lot / outhistory=summary;
run;
```

The data set SUMMARY contains variables named LOT, WIDTHM, WIDTHR, WIDTHN, DIAMTERM, DIAMTERR, and DIAMTERN.

Additionally, the following variables, if specified, are included:

- BY variables
- *block-variables*
- *symbol-variable*
- ID variables
- _PHASE_ (if the OUTPHASE= option is specified)

For an example of an OUTHISTORY= data set, see “Saving Summary Statistics” on page 1225.

OUTTABLE= Data Set

The OUTTABLE= data set saves subgroup summary statistics, control limits, and related information. The following variables are saved:

Variable	Description
ALPHA	probability (α) of exceeding control limits
EXLIM	control limit exceeded on median chart
EXLIMR	control limit exceeded on range chart
LCLM	lower control limit for median
LCLR	lower control limit for range
LIMITN	nominal sample size associated with the control limits
MEAN	estimate of process mean (\bar{M} , \bar{M} , \bar{X} , or μ_0)
R	average range
SIGMAS	multiple (k) of the standard error associated with control limits
<i>subgroup</i>	values of the subgroup variable
SUBM	subgroup median
SUBN	subgroup sample size
SUBR	subgroup range
TESTS	tests for special causes signaled on median chart
TESTS2	tests for special causes signaled on range chart
UCLM	upper control limit for mean
UCLR	upper control limit for range
VAR	<i>process</i> specified in the MRCHART statement

In addition, the following variables, if specified, are included:

- BY variables
- *block-variables*
- *symbol-variable*
- ID variables
- `_PHASE_` (if the `READPHASES=` option is specified)

Notes:

1. Either the variable `_ALPHA_` or the variable `_SIGMAS_` is saved depending on how the control limits are defined (with the `ALPHA=` or `SIGMAS=` options, respectively, or with the corresponding variables in a `LIMITS=` data set).
2. The variable `_TESTS_` is saved if you specify the `TESTS=` option. The k^{th} character of a value of `_TESTS_` is k if Test k is positive at that subgroup. For example, if you request all eight tests and Tests 2 and 8 are positive for a given subgroup, the value of `_TESTS_` has a 2 for the second character, an 8 for the eighth character, and blanks for the other six characters.
3. The variable `_TESTS2_` is saved if you specify the `TESTS2=` option. The k^{th} character of a value of `_TESTS2_` is k if Test k is positive at that subgroup.
4. The variables `_VAR_`, `_EXLIM_`, `_EXLIMR_`, `_TESTS_`, and `_TESTS2_` are character variables of length 8. The variable `_PHASE_` is a character variable of length 16. All other variables are numeric.

For an example of an `OUTTABLE=` data set, see “Saving Control Limits” on page 1226.

ODS Tables

The following table summarizes the ODS tables that you can request with the `MRCHART` statement.

Table 36.24. ODS Tables Produced with the `MRCHART` Statement

Table Name	Description	Options
MRCHART	median and R chart summary statistics	TABLE, TABLEALL, TABLEC, TABLEID, TABLELEG, TABLEOUT, TABLETESTS
Tests	descriptions of tests for special causes requested with the <code>TESTS=</code> option for which at least one positive signal is found	TABLEALL, TABLELEG

Input Data Sets

DATA= Data Set

You can read raw data (process measurements) from a DATA= data set specified in the PROC SHEWHART statement. Each *process* specified in the MRCHART statement must be a SAS variable in the DATA= data set. This variable provides measurements that must be grouped into subgroup samples indexed by the values of the *subgroup-variable*. The *subgroup-variable*, which is specified in the MRCHART statement, must also be a SAS variable in the DATA= data set. Each observation in a DATA= data set must contain a value for each *process* and a value for the *subgroup-variable*. If the t^{th} subgroup contains n_i items, there should be n_i consecutive observations for which the value of the *subgroup-variable* is the index of the t^{th} subgroup. For example, if each subgroup contains five items and there are 30 subgroup samples, the DATA= data set should contain 150 observations.

Other variables that can be read from a DATA= data set include

- `_PHASE_` (if the READPHASES= option is specified)
- *block-variables*
- *symbol-variable*
- BY variables
- ID variables

By default, the SHEWHART procedure reads all of the observations in a DATA= data set. However, if the DATA= data set includes the variable `_PHASE_`, you can read selected groups of observations (referred to as *phases*) by specifying the READPHASES= option (for an example, see “Displaying Stratification in Phases” on page 1689).

For an example of a DATA= data set, see “Creating Charts for Medians and Ranges from Raw Data” on page 1220.

LIMITS= Data Set

You can read preestablished control limits (or parameters from which the control limits can be calculated) from a LIMITS= data set specified in the PROC SHEWHART statement. For example, the following statements read control limit information from the data set CONLIMS:*

```
proc shewhart data=info limits=conlims;
  mrchart weight*batch;
run;
```

The LIMITS= data set can be an OUTLIMITS= data set that was created in a previous run of the SHEWHART procedure. Such data sets always contain the variables required for a LIMITS= data set. The LIMITS= data set can also be created directly using a DATA step. When you create a LIMITS= data set, you must provide one of the following:

*In Release 6.09 and in earlier releases, it is necessary to specify the READLIMITS option.

- the variables `_LCLM_`, `_MEAN_`, `_UCLM_`, `_LCLR_`, `_R_`, and `_UCLR_`, which specify the control limits directly
- the variables `_MEAN_` and `_STDDEV_`, which are used to calculate the control limits according to the equations in Table 36.22 on page 1243

In addition, note the following:

- The variables `_VAR_` and `_SUBGRP_` are required. These must be character variables of length 8.
- The variable `_INDEX_` is required if you specify the `READINDEX=` option; this must be a character variable of length 16.
- The variables `_LIMITN_`, `_SIGMAS_` (or `_ALPHA_`), and `_TYPE_` are optional, but they are recommended to maintain a complete set of control limit information. The variable `_TYPE_` must be a character variable of length 8; valid values are `ESTIMATE`, `STANDARD`, `STDMU`, and `STDSIGMA`.
- BY variables are required if specified with a BY statement.

For an example, see “Reading Preestablished Control Limits” on page 1229.

HISTORY= Data Set

You can read subgroup summary statistics from a `HISTORY=` data set specified in the `PROC SHEWHART` statement. This allows you to reuse `OUTHISTORY=` data sets that have been created in previous runs of the `SHEWHART` procedures or to read output data sets created with SAS summarization procedures, such as `PROC UNIVARIATE`.

A `HISTORY=` data set used with the `MRCHART` statement must contain the following variables:

- the *subgroup-variable*
- a subgroup mean variable for each *process*
- a subgroup median variable for each *process*
- a subgroup range variable for each *process*
- a subgroup sample size variable for each *process*

The names of the subgroup mean, subgroup median, subgroup range, and subgroup sample size variables must be the *process* name concatenated with the special suffix characters *X*, *M*, *R*, and *N*, respectively. You must provide the subgroup mean variable only if you specify the `MEDCENTRAL=AVGMEAN` option.

For example, consider the following statements:

```
proc shewhart history=summary;
  mrchart (weight yldstren)*batch / medcentral=avgmean;
run;
```

The data set `SUMMARY` must include the variables `BATCH`, `WEIGHTX`, `WEIGHTM`, `WEIGHTR`, `WEIGHTN`, `YLDSRENX`, `YLDSRENM`, `YLDSRENR`, and `YLDSRENN`.

Note that if you specify a *process* name that contains eight characters, the names of the summary variables must be formed from the first four characters and the last three characters of the *process* name, suffixed with the appropriate character.

Other variables that can be read from a HISTORY= data set include

- `_PHASE_` (if the READPHASES= option is specified)
- *block-variables*
- *symbol-variable*
- BY variables
- ID variables

By default, the SHEWHART procedure reads all the observations in a HISTORY= data set. However, if the data set includes the variable `_PHASE_`, you can read selected groups of observations (referred to as *phases*) by specifying the READPHASES= option (see “Displaying Stratification in Phases” on page 1689 for an example).

For an example of a HISTORY= data set, see “Creating Charts for Medians and Ranges from Summary Data” on page 1222.

TABLE= Data Set

You can read summary statistics and control limits from a TABLE= data set specified in the PROC SHEWHART statement. This enables you to reuse an OUTTABLE= data set created in a previous run of the SHEWHART procedure or to read data sets created by other SAS procedures. Because the SHEWHART procedure simply displays the information in a TABLE= data set, you can use TABLE= data sets to create specialized control charts. Examples are provided in Chapter 49, “Specialized Control Charts.”

The following table lists the variables required in a TABLE= data set used with the MRCHART statement:

Table 36.25. Variables Required in a TABLE= Data Set

Variable	Description
<code>_LCLM_</code>	lower control limit for median
<code>_LCLR_</code>	lower control limit for range
<code>_LIMITN_</code>	nominal sample size associated with the control limits
<code>_MEAN_</code>	process mean
<code>_R_</code>	average range
<i>subgroup-variable</i>	values of the <i>subgroup-variable</i>
<code>_SUBM_</code>	subgroup median
<code>_SUBN_</code>	subgroup sample size
<code>_SUBR_</code>	subgroup range
<code>_UCLM_</code>	upper control limit for median
<code>_UCLR_</code>	upper control limit for range

Other variables that can be read from a TABLE= data set include

- *block-variables*
- *symbol-variable*
- BY variables
- ID variables
- _PHASE_ (if the READPHASES= option is specified). This variable must be a character variable of length 16.
- _TESTS_ (if the TESTS= option is specified). This variable is used to flag tests for special causes for subgroup medians and must be a character variable of length 8.
- _TESTS2_ (if the TESTS2= option is specified). This variable is used to flag tests for special causes for subgroup ranges and must be a character variable of length 8.
- _VAR_. This variable is required if more than one *process* is specified or if the data set contains information for more than one *process*. This variable must be a character variable of length 8.

For an example of a TABLE= data set, see “Saving Control Limits” on page 1226.

Methods for Estimating the Standard Deviation

When control limits are determined from the input data, two methods are available for estimating the process standard deviation σ .

Default Method

The default estimate for σ is

$$\hat{\sigma} = \frac{R_1/d_2(n_1) + \cdots + R_N/d_2(n_N)}{N}$$

where N is the number of subgroups for which $n_i \geq 2$, and R_i is the sample range of the observations x_{i1}, \dots, x_{in_i} in the i^{th} subgroup.

A subgroup range R_i is included in the calculation only if $n_i \geq 2$. The unbiasing factor $d_2(n_i)$ is defined so that, if the observations are normally distributed, the expected value of R_i is equal to $d_2(n_i)\sigma$. Thus, $\hat{\sigma}$ is the unweighted average of N unbiased estimates of σ . This method is described in the *ASTM Manual on Presentation of Data and Control Chart Analysis* (1976).

MVLUE Method

If you specify SMETHOD=MVLUE, a minimum variance linear unbiased estimate (MVLUE) is computed for σ . Refer to Burr (1969, 1976) and Nelson (1989, 1994). The MVLUE is a weighted average of N unbiased estimates of σ of the form $R_i/d_2(n_i)$, and it is computed as

$$\hat{\sigma} = \frac{f_1 R_1 / d_2(n_1) + \cdots + f_N R_N / d_2(n_N)}{f_1 + \cdots + f_N}$$

where

$$f_i = \frac{[d_2(n_i)]^2}{[d_3(n_i)]^2}$$

A subgroup range R_i is included in the calculation only if $n_i \geq 2$, and N is the number of subgroups for which $n_i \geq 2$. The MVLUE assigns greater weight to estimates of σ from subgroups with larger sample sizes, and it is intended for situations where the subgroup sample sizes vary. If the subgroup sample sizes are constant, the MVLUE reduces to the default estimate.

See Example 36.1 on page 1253 for illustrations of the default and MVLUE methods.

Axis Labels

You can specify axis labels by assigning labels to particular variables in the input data set, as summarized in the following table:

Axis	Input Data Set	Variable
Horizontal	all	<i>subgroup-variable</i>
Vertical (median chart)	DATA=	<i>process</i>
Vertical (median chart)	HISTORY=	subgroup median variable
Vertical (median chart)	TABLE=	_SUBMED_

You can specify distinct labels for the vertical axes of the median and R charts by breaking the vertical axis into two parts with a split character. Specify the split character with the `SPLIT=` option. The first part labels the vertical axis of the median chart, and the second part labels the vertical axis of the R chart.

For an example, see Example 36.2 on page 1257.

Missing Values

An observation read from a `DATA=`, `HISTORY=`, or `TABLE=` data set is not analyzed if the value of the subgroup variable is missing. For a particular process variable, an observation read from a `DATA=` data set is not analyzed if the value of the process variable is missing. Missing values of process variables generally lead to unequal subgroup sample sizes. For a particular process variable, an observation read from a `HISTORY=` or `TABLE=` data set is not analyzed if the values of any of the corresponding summary variables are missing.

Examples

This section provides advanced examples of the MRCHART statement.

Example 36.1. Working with Unequal Subgroup Sample Sizes

A brewery monitors its bottling process to ensure that each bottle is filled with the proper amount of beer. The following data set contains the amount of beer recorded in fluid ounces for 23 batches:

See SHWMR2
in the SAS/QC
Sample Library

```

data beer;
  input batch size @;
  do i=1 to size;
    input amount @@;
    output;
  end;
  drop i size;
  label batch = 'Batch Number';
  datalines;
1  5  12.01 11.97 11.93 11.98 12.00
2  5  11.88 11.98 11.93 12.03 11.92
3  5  11.93 11.99 12.00 12.03 11.95
4  5  11.98 11.94 12.02 11.90 11.97
5  5  12.02 12.02 11.98 12.04 11.90
6  4  11.98 11.98 12.00 11.93
7  5  11.93 11.95 12.02 11.91 12.03
8  5  12.00 11.98 12.02 11.89 12.01
9  5  11.98 11.93 11.99 12.02 11.91
10 5  11.97 12.02 12.05 12.01 11.97
11 5  12.02 12.01 11.97 12.02 11.94
12 5  11.93 11.83 11.99 12.02 12.01
13 5  12.01 11.98 11.94 12.04 12.01
14 5  11.98 11.96 12.02 12.00 12.00
15 5  11.97 11.99 12.03 11.95 11.96
16 5  11.99 11.95 11.96 12.03 12.01
17 4  11.99 11.97 12.03 12.01
18 5  11.94 11.96 11.98 12.03 11.97
19 5  11.97 11.87 11.90 12.01 11.95
20 5  11.96 11.94 11.96 11.98 12.05
21 3  12.06 12.07 11.98
22 5  12.01 11.98 11.96 11.97 12.00
23 5  12.00 12.02 12.03 11.99 11.96
;

```

A batch is regarded as a rational subgroup. Five bottles of beer are supposed to be tested in each batch. However, in batch 6 and batch 17 only four bottles are tested, and in batch 21 only three bottles are tested. The following statements request median and range charts, shown in Output 36.1.1, for the beer amounts:


Part 9. The CAPABILITY Procedure

```

title 'Median and Range Charts for Beer Amount';
symbol v=dot;
proc shewhart data=beer;
  mrchart amount*batch;
run;

```

Output 36.1.1. Median and Range Charts with Varying Sample Sizes



Since none of the subgroup medians or subgroup ranges fall outside their respective control limits, you can conclude that the process is in control.

Note that the central line on the range chart and the control limits on both charts vary with the subgroup sample size. The subgroup sample size legend displays the minimum and maximum subgroup sample sizes.

The SHEWHART procedure provides various options for working with unequal subgroup sample sizes. For example, you can use the LIMITN= option to specify a fixed (nominal) sample size for the control limits, as illustrated by the following statements:


```

title 'Median and Range Charts for Beer Amount';
symbol v=dot;
proc shewhart data=beer;
  mrchart amount*batch / limitn=5;
run;

```

The resulting charts are shown in Output 36.1.2.


Output 36.1.2. Control Limits Based on Fixed Sample Size



Note that the points displayed on the chart are those corresponding to subgroups whose sample size matches the nominal sample size (five) specified with the LIM-ITN= option. Points are not plotted for batches 6, 17, and 21. To display points for all subgroups (regardless of subgroup sample size), specify the ALLN option. The following statements produce the charts shown in Output 36.1.3:

```
proc shewhart data=beer;
  mrchart amount*batch / limitn=5
                        alln
                        nmarkers;
run;
```

The NMARKERS option requests special symbols that identify points for which the subgroup sample size differs from the nominal sample size. In Output 36.1.3, the median amount for batch 21 exceeds the upper control limits, indicating that the process is not in control. This illustrates the approximate nature of fixed control limits used with subgroup samples of varying sizes.

Output 36.1.3. Displaying All Subgroups Regardless of Sample Size

You can use the SMETHOD= option to determine how the process standard deviation σ is to be estimated when the subgroup sample sizes vary. The default method computes σ as an unweighted average of subgroup estimates of σ . The MVLUE method assigns greater weight to estimates of σ from subgroups with larger sample sizes. If the subgroup sample sizes are constant, the MVLUE method reduces to the NOWEIGHT method.

For details, see “Methods for Estimating the Standard Deviation” on page 1251. The following statements estimate σ using both methods:

```
proc shewhart data=beer;
  mrchart amount*batch / outindex = 'Default'
                        outlimits = blim1
                        nochart;
  mrchart amount*batch / smethod  = mvlue
                        outindex  = 'MVLUE'
                        outlimits  = blim2
                        nochart;
run;

data blimits;
  set blim1 blim2;
run;
```

The estimates are saved as values of the variable `_STDDEV_` in the data set `BLIMITS`, which is listed in Output 36.1.4. The bookkeeping variable `_INDEX_` identifies the estimate.

Output 36.1.4. The Data Set BLIMITS

The Data Set BLIMITS													
	—			—	—						—		
	S			L	—	S					S		
	U	I		I	A	I					T		
—	B	N	T	M	L	G	L	M	U	L	U	D	
V	G	D	Y	I	P	M	C	E	C	C	C	D	
A	R	E	P	T	H	A	L	A	L	L	—	L	E
R	P	X	E	N	A	S	M	N	M	R	R	R	V
—	—	—	—	—	—	—	—	—	—	—	—	—	—
amount	batch	Default	ESTIMATE	V	V	3	V	11.9856	V	V	V	V	0.043938
amount	batch	MVLUE	ESTIMATE	V	V	3	V	11.9856	V	V	V	V	0.044004

In the data set BLIMITS, the variables `_LIMITN_`, `_ALPHA_`, `_LCLM_`, `_UCLM_`, `_LCLR_`, `_R_`, and `_UCLR_` have been assigned the special missing value V. This indicates that the quantities represented by these variables vary with the subgroup sample size.

Example 36.2. Specifying Axis Labels

This example illustrates various methods for specifying axis labels and other axis features for median and range charts. For further details, see “Labeling Axes” on page 1719.

See SHWMR3
in the SAS/QC
Sample Library

The charts in Figure 36.2 on page 1222, which are based on the data set DETERGNT introduced in the “Getting Started” section on page 1220, display default labels for the horizontal and vertical axes. You can specify axis labels by associating labels with the *process* and *subgroup* variables as illustrated by the following statements:


```

title 'Median and Range Charts for Detergent Box Weight';
symbol v=dot;
proc shewhart data=detergnt;
  mrchart weight*lot / split = '//';
  label lot      = 'Detergent Lot Number'
        weight = 'Median (pounds)/Range';
run;

```

The charts are shown in Output 36.2.1. The horizontal axis label is the label associated with the *subgroup-variable* LOT. The vertical axis label for the median chart, referred to as the primary vertical axis label, is the first portion of the label associated with the *process* variable WEIGHT, up to but not including the split character, which is specified with the `SPLIT=` option. The vertical axis label for the range chart, referred to as the secondary vertical axis label, is the second portion of the label associated with WEIGHT.

Output 36.2.1. Customized Axis Labels Using Variable Labels



When the input data set is a HISTORY= data set, the vertical axis labels are determined by the label associated with the subgroup median variable. This is illustrated by the following statements, which use the data set DETSUM introduced on page 1222:

```

title 'Median and Range Charts for Detergent Box Weight';
symbol v=dot;
proc shewhart history=detsum;
  mrchart weight*lot / split = '//';
  label lot      = 'Detergent Lot Number'
        weightm = 'Median (pounds)/Range';
run;

```

The charts are identical to those in Output 36.2.1.

When the input data set is a TABLE= data set, the vertical axis labels are determined by the label associated with the subgroup median variable _SUBMED_. This is illustrated by the following statements, which use the data set DTABLE introduced in Figure 36.7 on page 1228:

```

title 'Median and Range Charts for Detergent Box Weight';
symbol v=dot;
proc shewhart table=dtable;
  mrchart weight*lot / split = '//';
  label lot      = 'Detergent Lot Number'
        _submed_ = 'Median (pounds)/Range';
run;

```

The charts are identical to those in Output 36.2.1.

When you are creating charts on graphics devices, you can use AXIS statements to enhance the appearance of the axes. This method is illustrated by the following statements:

```

title 'Median and Range Charts for Detergent Box Weight';
symbol v=dot;


proc shewhart data=detergnt;
  mrchart weight*lot / haxis = axis1
                    vaxis = axis2
                    vaxis2 = axis3 ;

  axis1 label = ( color=red 'Detergent Lot Number' );
  axis2 label = ( color=blue 'Median Weight' j=c 'of Sample' );
  axis3 label = ( color=blue 'Range of' j=c 'Sample' );
run;

```

The charts are shown in Output 36.2.2.

Output 36.2.2. Customized Axis Labels Using AXIS Statements



You can use AXIS statements to customize a variety of axis features. For details, see *SAS/GRAPH Software: Reference*.

The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *SAS/QC[®] User's Guide, Version 8*, Cary, NC: SAS Institute Inc., 1999. 1994 pp.

SAS/QC[®] User's Guide, Version 8

Copyright © 1999 SAS Institute Inc., Cary, NC, USA.

ISBN 1-58025-493-4

All rights reserved. Printed in the United States of America. No part of this publication may be reproduced, stored in a retrieval system, or transmitted, by any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

U.S. Government Restricted Rights Notice. Use, duplication, or disclosure of the software by the government is subject to restrictions as set forth in FAR 52.227-19 Commercial Computer Software-Restricted Rights (June 1987).

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999

SAS[®] and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute in the USA and other countries. [®] indicates USA registration.

IBM[®], ACF/VTAM[®], AIX[®], APPN[®], MVS/ESA[®], OS/2[®], OS/390[®], VM/ESA[®], and VTAM[®] are registered trademarks or trademarks of International Business Machines Corporation. [®] indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The Institute is a private company devoted to the support and further development of its software and related services.